

Simulation d'une transmission RFID

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)


d'autres pdf, sur différents sujets :

[liste des PDE](#)

La « RFID » (radio frequency identification) est le domaine du « sans contact » : des données (un identifiant) sont transmises d'un « tag » vers un lecteur. Ce document traite une transmission dans la bande 125 kHz, où la proximité entre le tag et le récepteur de données est de l'ordre du cm. Par simplicité, on ne considère que la lecture du tag et non pas l'écriture dans le tag. Avec des tags passifs, le lecteur (ou base) transmet la puissance pour alimenter l'électronique interne du tag. Nous allons également analyser cette action.

1) Principe, introduction

En RFID, la propagation des données d'une part, et de la puissance d'autre part, se fait par couplage magnétique : une source, placée dans la base, alimente un bobinage que l'on appellera primaire, et un autre bobinage, couplé magnétiquement, alimente à son tour l'électronique du tag. Les 2 bobinages forment un transformateur à air. La donnée numérique ouvre et ferme un interrupteur ce qui provoque des appels de courant au primaire. Une électronique placée dans la base détecte ce courant pour reformer les données.



Pour proposer une analyse proche de la réalité, des selfs ANT125K-20 de Eccel Technology Ltd (référence RS : 126-2187) ont été utilisées. (Diamètre externe 2 cm, interne 1 cm, épaisseur 2 mm). C'est le genre de bobines que l'on retrouve dans les « key fob ».



Bobine utilisée et key fob

Une mesure préalable (moyenne sur 5 bobines) a donné une résistance série DC de 9,7 Ω .

L'inductance propre, dans un milieu non magnétique : 794 μH .

Ces mesures d'inductance ont été faites à 100 Hz, 1 kHz et 10 kHz, au pont LCR400 : pas de dispersion significative. La résistance passe à 9,9 Ω à 10 kHz, par effet de peau et/ou de proximité.

On est un peu éloigné du document constructeur qui donne 750 μH et 7,5 Ω DC.

La théorie des transformateurs est connue, surtout quand les enroulements primaire et secondaire forment un couplage parfait, obtenu notamment par l'utilisation de matériaux ferromagnétiques. Ici, la situation est très différente, car le primaire et le secondaire sont distants l'un de l'autre, et il n'y a pas de circuit magnétique. De plus, à l'utilisation, l'espace entre les 2 bobinages varie de l'infini (utilisateur loin de la base) vers quelques mm quand le tag est en quasi contact avec la base.

Ce document commence par évaluer le coefficient de couplage du transformateur à air ainsi formé, valeur inconnue. Seule, la proximité proche est intéressante à analyser. Nous fixerons la valeur nominale entre les 2 bobines à 1 cm.

Il a été choisi que la bobine de la base serait identique. C'est ce qu'on trouve parfois dans les lecteurs de badges de porte d'entrée. Ici, l'intérêt de placer 2 bobines identiques est de simplifier l'étude.

1) 2 bobines à proximité forment un transformateur à air

Pour faire une campagne de caractérisation, 2 bobines face à face séparées par une cale (non magnétique) ont été installées. Cette cale est réalisée à partir de petits blocs de « post it ».

En retirant des feuilles, on peut paramétrer l'espace entre les 2 bobines. Il y a eu 7 essais : 12, 10, 8, 6, 4, 2, 0 mm.




Photo montrant 2 bobines face à face. Ici, l'espace est de 10 mm.

1.a) Détermination de l'inductance mutuelle

Rappel :

On appelle **inductance propre**, (ou coefficient d'auto inductance) : de même pour L_2 .

$$L_1 = \frac{\text{Flux créé par la bobine B1 à travers elle même}}{\text{courant dans la bobine B1}} \quad (H)$$

La **mutuelle inductance**, (ou coefficient de mutuelle inductance) est :

$$M = \frac{\text{Flux créé par la bobine B1 à travers la bobine B2}}{\text{courant dans la bobine B1}} \quad (H)$$

(et réciproquement)

M peut varier entre 0 (L_1 et L_2 séparées magnétiquement) et $\sqrt{L_1 L_2}$ (couplage maximum)

Prenons le cas général où la bobine 1 a une inductance propre L_1 , et la bobine 2, L_2 .

Une façon de mesurer l'inductance mutuelle entre 2 inductances est de les câbler en série. 2 cas peuvent se produire :


Champ additif (les ampères-tours sont dans le même sens) :

$$L_a = L_1 + L_2 + 2M$$

Champ soustractif (les ampères-tours sont en sens opposé) :

$$L_s = L_1 + L_2 - 2M.$$

On déduit : $L_a - L_s = 4 M$ d'où $M = (L_a - L_s) / 4$




Le sens des spires de L_2 diffère entre (a) et (b).

Expérimentalement, pour inverser les ampères tours, on peut retourner physiquement une self. Mais il faut alors la placer dans la même position, ce qui, sans artifice de guidage ou de positionnement est délicat. Il a été choisi de permuter les fils d'une self, ce qui permet de ne pas déplacer les bobines pour mesurer L_a et L_s . On s'assure également qu'il n'y a aucune masse métallique à proximité.

Le tableau ci-dessous récapitule les mesures :

Espace (mm)	L_a (mH)	L_s (mH)	$M = \frac{L_a - L_s}{4}$ (mH)	$K = \frac{M}{\sqrt{L_1 \cdot L_2}}$
0	2,58	0,546	0,5085	0,64
2	2,2	0,986	0,3035	0,38
4	1,97	1,216	0,1885	0,24
6	1,83	1,356	0,1185	0,15
8	1,746	1,44	0,0765	0,1
10	1,7	1,47	0,0575	0,08
12	1,683	1,505	0,0445	0,06



M (en mH) = f (e) (en mm)


Interprétation :

On vérifie que plus on éloigne les 2 selfs, plus on réduit la mutuelle inductance. Avec L_1, L_2 très éloignées entre elles, donc non couplées magnétiquement, la mise en série donne une inductance résultante $L = L_1 + L_2 = 1,588 \text{ mH}$ (on retrouve $M = 0$).

Remarque : dans notre cas particulier où $L_1 = L_2$, on peut exploiter une relation plus simple : $L_a = 2 L_1 + 2 M$, d'où $M = (L_a - 2 L_1) / 2$ ou alors : $L_s = 2 L_1 - 2 M$ d'où $M = (2 L_1 - L_s) / 2$. La campagne de caractérisation nécessite alors 7 mesures au lieu de 14.

1.b) Le transformateur à air.

Rappel : Un transformateur sans modèle de noyau magnétique est défini par les valeurs de L_1, L_2, K , ce qui se traduit par le modèle donné ci-contre :



$$K = \frac{M}{\sqrt{L_1 \cdot L_2}}$$


Le tableau précédent montre, en dernière colonne, que le coefficient de couplage chute avec la distance. Dans notre meilleur des cas de couplage (bobines empilées comme 2 pièces de monnaie), le coefficient $K = 0,64$. Il y a donc une partie du flux créé par L_1 qui ne traverse pas L_2 , malgré le quasi collage des faces. C'est la conséquence de l'absence de circuit magnétique pour canaliser toutes les lignes de champ.

Dans notre configuration (mêmes bobinages primaire et secondaire) on peut écrire la relation $\vec{V}_2 = -K\vec{V}_1$. Avec un nombre de spires différent, en posant $m = N_2/N_1$, nous aurions $K = M/(mL_1) = m M/L_2$, et par conséquence, $\vec{V}_2 = -K m \vec{V}_1$.

2) Exploitation du transformateur à air pour transmettre la puissance

Travaillons avec un espacement de 10 mm, c'est-à-dire $K = 0,08$:

En restant dans le domaine « basse fréquence », on peut adopter classiquement le modèle du transformateur suivant :



Avec : $r_p = r_s = 9,7 \Omega$. $K = 0,08$ $L_1 = 794 \mu\text{H}$.

Rappel : $N_2 = N_1$ d'où $m = 1$

On déduit :

$$\sigma = 1 - K^2 = 1 - 0,08^2 = 0,9936,$$

l'inductance de fuite, calculée par PSpice : $L_f = \sigma L_2 \approx 790 \mu\text{H}$.

On a, à 125 kHz, $L_f \omega = 620 \Omega$.

2.a) Simulation du transformateur à air

Le primaire reçoit une tension sinusoïdale, et connectons une résistance R_L sur le secondaire. Vérifions le faible rendement en puissance.

On fixe arbitrairement 2 valeurs à R_L : 100Ω et $10 \text{ k}\Omega$.

On relève la réponse **harmonique** (dans une très large bande de fréquence pour mieux centrer les courbes mais le modèle est mis en défaut au-delà de quelques MHz, quand les condensateurs parasites interviennent).





Schéma simulé


bobines faiblement couplées
 * fichier transfoRFID.cir
 * circuit :
 * génération du 125 kHz
 V125 E 0 AC=1
 * primaire : emet la puissance
 Rp E i 9.7 ;
 L1 i 0 794uH ; ANT125K-20 de Eccel
 Ktransfo L1 L2 0.08
 L2 2 0 794uH ; bobine identique
 Rs 2 A 9.7
 .param Rcharge = 10k
 RL A 0 {Rcharge} ; modélise conso
 .STEP param Rcharge list 10k 100
 .AC DEC 1000 1 1000MEG
 .probe
 .end
Netlist, prête à simuler

Comportement en fréquence


On affiche $V(A)$, pour avoir la réponse harmonique en **tension**.
 Pour $R_L = 10\text{ k}\Omega$, les courants primaire et secondaire sont faibles et les chutes de tension dans r_p , r_s sont négligeables : il existe un plateau dans lequel on a la relation $\vec{V}_2 = -K \vec{V}_1$, avec $K = 0,08$. Cette valeur est maintenue à 125 kHz.
 Pour $R_L = 100\ \Omega$, la bande de fréquence d'utilisation en tension du transformateur est très réduite.



On affiche $I(L2) / I(L1)$, pour avoir la réponse harmonique en **courant**.
 La réponse harmonique en courant montre qu'au-delà d'une certaine fréquence, on a la relation : $\vec{I}_2 = -K \vec{I}_1$, ce qui est le fonctionnement normal du transformateur parfait (pour $m = 1$).




La réponse harmonique en **puissance**, montre qu'en basses fréquences, la source débite 103,09 mW. C'est en fait la dissipation dans r_s , que l'on vérifie par la loi de Joule : $P = U_{ef}^2 / r_s = 1 / 9,7 = 0,10309\text{ W}$.
 La dissipation dans R_L dépend de la présence simultanée de courant et tension, ce qui explique la bande étroite.



Faisons le bilan des puissances à la fréquence 125 kHz :
 Pour $R_L = 100\ \Omega$: la source débite 1,61 mW, R_L dissipe 1,618 μW : le rendement est de 0,1 %.
 Pour $R_L = 10\text{ k}\Omega$: la source débite 1,61 mW, R_L dissipe 0,636 μW : le rendement est de 0,04 %.

Observons le **rendement** fonction de la fréquence :
 Avec cette valeur de K , on a, au maximum, le rendement en puissance de 0,29 % pour $R_L = 100\ \Omega$ (à 22,4 kHz) et 0,32 % pour $R_L = 10\text{ k}\Omega$ (à 2 MHz).
 A 125 kHz, on retrouve les valeurs de rendement : 0,1 % pour $R_L = 100\ \Omega$ et 0,04 % pour $R_L = 10\text{ k}\Omega$.



La conclusion de ces premières simulations est de constater le très faible transfert de puissance dans un transformateur à air. Une amélioration consiste à placer des condensateurs pour faire des résonateurs.

2) L'apport de condensateurs pour travailler à la résonance

2.a) Principe

Pour réduire l'impact de la perte de tension entre primaire et secondaire, on peut exploiter le phénomène de résonance : en connectant sur le secondaire un condensateur, on peut bénéficier, à une fréquence particulière, d'un phénomène de « surtension ». La tension secondaire est d'autant plus élevée que le facteur de qualité est élevé.

La lecture des données en provenance du badge nécessite un système de captage de courant primaire. On fait appel à la tension aux bornes d'un condensateur, placé en série avec le primaire. Également, de choisir une valeur pour être à la résonance permet d'augmenter la tension à ces bornes.

Simplifions le schéma de cette installation (voir ci-dessous) en posant : $L_1 = L_2 = L$, $C_1 = C_2 = C$. En négligeant dans un premier temps les résistances, il apparaît 2 circuits LC, formant 2 oscillateurs couplés par induction mutuelle.

L'étude de ce montage montre 2 fréquences de résonance : $\frac{1}{2\pi\sqrt{(L+M)C}}$ et $\frac{1}{2\pi\sqrt{(L-M)C}}$.

- Ces 2 fréquences sont confondues à $M = 0$, c'est-à-dire les 2 bobines éloignées, les 2 oscillateurs résonant avec leur propre LC.
- Avec couplage, ces 2 pics de résonance vont être présents sur la réponse harmonique de $V(A)/V(E)$, à des valeurs d'autant plus éloignées que les bobines sont proches. C'est ce que nous allons vérifier par la simulation.
- La contribution des résistances est de réduire le facteur de qualité et de maintenir à une valeur finie la tension secondaire à la résonance.

Application numérique : on applique $LC\omega^2 = 1$. A 125 kHz, avec $L = 794 \mu\text{H}$, cela donne $C = 2,042 \text{ nF}$, valeur que l'on reportera sur C_1 et C_2 .

2.b) Simulation, signal sinusoïdal

On analyse la conséquence du coefficient de couplage K (et donc de la distance entre le lecteur et le tag) sur le rapport de tension $V(A)/V(E)$.

On prend les 7 valeurs de K de l'expérience de départ. Entre 2 résonances, on relève la valeur minimale du transfert de tension.

On conserve une résistance $R_L = 10 \text{ k}\Omega$.




Schéma simulé

En pratique, la masse n'est pas commune entre le primaire et le secondaire. Il y a un point commun pour faciliter la simulation.

bobines faiblement couplées et résonance

* fichier transfoRFID_C.cir

* circuit :

* génération du 125 kHz

V125 E 0 SIN (0 1.414 125k) AC=1

* primaire : émet la puissance

Rp E i 9.7 ;

L1 i ii 794uH ; ANT125K-20 de Eccel

Cpri ii 0 2.042n

.param K=0.08

Ktransfo L1 L2 {K}

L2 2 0 794uH ; bobine identique

Rs 2 A 9.7

Csec A 0 2.042n

.param Rcharge = 10k

RL A 0 {Rcharge} ; modélise conso

.STEP param K list 0.06 0.08 0.1 0.15

+ 0.24 0.38 0.64


.AC DEC 1000 10 100MEG

.TRAN 0.1u 1m 0.4m 0.1u

.probe

.end

Netlist, prête à simuler



V(A)/V(E) : 2 fréquences de résonance liées à l'espacement entre les 2 bobines

Au curseur, on mesure, après cette simulation :

	K = 0,06	K = 0,08	K = 0,1	K = 0,15	K = 0,24	K = 0,38	K = 0,64
Fréquence de résonance	123,9 kHz	121,7 kHz	120,4 kHz	117,3 kHz	112,6 kHz	106,6 kHz	97,7 kHz
	126,5 kHz	128,9 kHz	130,7 kHz	135 kHz	143,1 kHz	108,3 kHz	208,1 kHz
Transfert minimum entre les résonances	12,4	10,5	9	6,3	4	2,51	1,37


Interressons-nous à notre cas de travail **K = 0,08** (espacement entre les bobines de 1 cm). Les 2 résonances sont suffisamment proches pour avoir, à 125 kHz, secondaire chargé par 10 kΩ, une amplification en tension de 10,5, alors qu'il y avait 0,08 sans condensateur. Le gain dans le transfert en tension est évident. Vérifions maintenant la « tenue en puissance » de ce montage.

Transfert de puissance

A 125 kHz, la source émet 16,4 mW, R_L en récupère 11 mW. L'écart est la puissance dissipée dans r_p (2,6 mW) et r_s (2,76 mW).

Rendement :

Le rendement passe par un maximum à 125 kHz, et vaut, pour K = 0,08 et R_L = 10 kΩ, 67%.




Puissances et rendement fonction de la fréquence

Reponse temporelle, à 125 kHz.

La source sinusoïdale a 1 V eff, afin de retrouver les chiffres déterminés en AC.

On recueille en V(A) 29,6 V càc soit 10,469 V eff. On retrouve bien la valeur d'amplification de 10,5. De même, le bilan des puissances est confirmé :

-AVGX(W(V125),32u)	16.416m
AVGX(W(RL),32u)	11.037m
AVGX(W(Rp),32u)	2.6116m
AVGX(W(Rs),32u)	2.7503m



**à 125 kHz : 1 V_{eff} en V(E) → 10,5 V_{eff} en V(A)
Dans ces conditions, la puissance qui passe entre la base et le tag est de 13,8 mW**


On conclut, que si l'électronique du tag est équivalente à R = 10 kΩ (ce qui représente un modèle simplifié de l'association de tous les constituants de récepteur), ce dernier, placé à 1 cm de la base, peut dissiper 11 mW, pour 1 V_{eff} généré par la base.

3) Le transfert des données

La puce du tag est donc alimentée en puissance grâce à la proximité de la base. Durant ce temps, le circuit génère son identifiant, c'est-à-dire une suite de 0 et 1 que la base doit récupérer. Il a été choisi une « modulation de charge » pour transmettre cette information binaire.

3.a) Principe : modulation de charge

On exploite la propriété que le courant primaire est lié au courant secondaire, même à faible couplage. On présente au secondaire une résistance qui change instantanément de valeur par la donnée numérique. Une façon simple de réaliser ceci est d'ouvrir et fermer un interrupteur électronique qui place sur le secondaire une résistance de shunt.




Principe de la communication RFID

Pour rendre plus facile la détection, le premier réflexe est de proposer une résistance de shunt de faible valeur, tel un court-circuit. Mais de court-circuiter ainsi le secondaire a comme conséquence d'arrêter le transfert de puissance Base → Tag. Si on suppose autant de « 0 » que de « 1 », (hypothèse d'équiprobabilité couramment admise), la puissance récupérée sera alors divisée par 2. Il y a donc un compromis entre sensibilité pour la lecture de la base, et transfert de puissance vers le tag. Ajoutons un autre argument : la réduction de la bande passante à faible résistance de charge, comme montré précédemment.

Un compromis est de placer $R_{shunt} = 10 \text{ k}\Omega$. Dans ces conditions, il apparaît une « résistance équivalente » de $20 \text{ k}\Omega$ qui consommerait la même puissance.


3.b) Schéma

L'interrupteur est simulé par le composant « Voltage-Controlled Switch », qui est une résistance commandée en tension. Sa valeur est r_{on} et r_{off} , respectivement quand on applique V_{ON} , V_{OFF} sur une tension de commande.



On ouvre et ferme l'interrupteur au rythme de l'information binaire.

On va analyser dans un premier temps la **démodulation**.



En régime sinusoïdal, la tension aux bornes de C_1 s'écrit $\frac{I_1}{jC\omega}$. On a vu que, par couplage magnétique, I_1 est lié à I_2 , et ce dernier est lié à la position de l'interrupteur : l'amplitude de la tension en C est donc liée à V_{com} . C'est une modulation ASK.

Un montage détecteur de crête (diode et condensateur) + résistance permet de réaliser une démodulation d'amplitude simpliste, en délivrant un signal démodulé grossièrement au nœud **R**.

Un filtre de lissage (R C passe bas) améliore l'information et délivre une tension exploitable en **F**.

Un condensateur coupe la composante continue, en faisant un filtre passe haut à 1,6 kHz (10 nF ; 10 kΩ).

Un étage amplificateur non inverseur sort en **S1**. En sortie le signal est bipolaire.

Un étage comparateur tout ou rien à base de LM111 reforme un signal logique en **S2**.

Pour la simulation, la commande qui ouvre et ferme l'interrupteur est une source de tension parfaite, modélisant l'électronique délivrant le flot binaire à transmettre, au travers d'un transistor qui commute R_{shunt} .

Remarquons que la source qui alimente le primaire du transformateur peut être sinusoïdale, ou, plus simplement, carrée à la même fréquence : d'une part c'est plus simple de générer un signal logique, et d'autre part, le fondamental traversera le transformateur, les harmoniques étant hors bande passante.

Il a été placé une source de type « PULSE », de fréquence 125 kHz et de niveau 0 V , 5 V. qui modélise un circuit intégré logique alimenté en 5 V. Le fondamental a pour amplitude $\frac{4}{\pi} 2,5$ V, soit 2,25 V eff.

3.c) Simulation en modulation d'amplitude

Pour vérifier le fonctionnement, nous allons choisir un signal logique volontairement lent : une fréquence bit de 10 kHz, soit un taux de modulation de 10 kbauds.

<pre> transfert de données par modulation de charge *fichier data_RFID.cir * circuit : .lib eval.lib *Alimentations : VDC V5 0 DC=5 Vplus Vp 0 DC=15 Vmoins Vm 0 DC=-15 * génération du 125 kHz V125 E 0 PULSE (0 5 0 20n 20n 3.98u 8u) * primaire : emet la puissance Rp E i 9.7 ; L1 i C 794uH ; ANT125K-20 de Eccel Cpri C 0 2.042n Ktransfo L1 L2 0.08 L2 2 0 794u ; bobine identique Rs 2 A 9.7 Csec A 0 2.042n RL A 0 10k ; charge équivalente *Dpuce A K D1N4148 *RL K 0 30k ; modélise conso puce *CF K 0 1n ; RC = 30 us Rshunt A B 10k ; quand le switch est ON Sintc B 0 Com 0 INTER .MODEL INTER Vswitch (Ron=0.1 Roff=1Meg Von=2.1 Voff=1.9) Vcom Com 0 PULSE (0 4 0m 20n 20n 199.98u 400u) </pre>	<pre> XDemod C R Demod XFiltre R F Filtre Xampli F S1 Vp Vm V5 Ampli Xcomp S1 S2 Vp Vm V5 Comp .SUBCKT Filtre in out * filtre de lissage Rfil in out 4.7k ; RC = 4,7 us Cfil out 0 1n .ENDS .SUBCKT Demod in out * démodulateur AM que le primaire reçoit Dred in out D1N4148 Cred out 0 6.8n ; RC = 462 us Rdec out 0 68k .ENDS .SUBCKT Comp in out Vplus Vmoins V5 XCOMP in 0 Vplus Vmoins out 0 LM111 Rpullup out V5 1k ; sortie sur collecteur .ENDS .SUBCKT Ampli in out Vplus Vmoins V5 Cliai in ep 10n ; coupe DC Rpulldown ep 0 10k R20 em 0 220 X1 ep em Vplus Vmoins out LF411 R22 em out 1Meg .ENDS .TRAN 20n 10m 9m 20n .probe .end </pre>
---	--


Netlist, prête à simuler

L'amplificateur à base de LF411 (eval.lib) monté avec 220 Ω , 1 M Ω , a un coefficient d'amplification de 4545, soit 73,15 dB.

Le comparateur à base de LM111 (eval.lib) délivre un signal entre 0 V et 5 V, par pull up de 1 k Ω .

- Comme souvent, il faut laisser un temps de simulation assez long pour que les résultats soient en régime permanent, c'est-à-dire hors de la contribution des conditions initiales. Dans le cas présent, on regarde après 9 ms de simulation.


- Sur cette simulation, le flot binaire censé représenter l'identifiant est 0, 1, 0, 1, 0, (200 μ s à l'état haut, 200 μ s à l'état bas).



Explications

Quand V_{com} est à l'état « 1 » (4 V à 9,2 ms) :

- l'interrupteur est fermé. La tension secondaire $V(A)$ « s'écroule », mais modérément, car, rappelons-le, R_{shunt} n'est pas un court-circuit. La charge équivalente présentée est de $5\text{ k}\Omega$.
- Les courants secondaire et primaire sont élevés. L'amplitude en C est plus grande, ce qui explique un potentiel élevé en R . On remarque que la démodulation est sommaire, avec un temps de réponse.



- en $V(F)$, on a une copie de $V(R)$, mais lissée.
- Par suite, cette tension, retirée de sa composante continue fait basculer $S1$ qui se dirige vers -15 V (le fort gain de l'amplificateur entraîne la saturation), lors de la traversée à 0 V .
- L'étage comparateur remet en forme pour délivrer un niveau bas (0 V) en $S2$.

Quand V_{com} est à l'état « 0 » (0 V) :

- l'interrupteur est ouvert. La tension secondaire remonte, le transformateur est chargé par $R_L = 10\text{ k}\Omega$.
- L'amplitude en $V(C)$ est plus faible, $S1$ monte, se dirige vers $+15\text{ V}$, $S2$ affiche un niveau haut.

Au bilan, **S2** est 0, 1, 0, 1, 0, comme la commande. L'information binaire émise par le TAG est donc récupérée par la base. On remarque un décalage entre V_{com} et **S2** de 60 μs .


La puissance fournie par le signal carré à 125 kHz est de **66 mW**. Celle émise par L_1 et captée par L_2 est de **45,7 mW**. Au bout de la chaîne, modélisée ici par $R_L = 20\text{ k}\Omega$, (qui se trouve en parallèle avec une résistance équivalente de 20 $\text{k}\Omega$), on dispose de **22 mW**.

4) Le transfert de puissance base \rightarrow tag

La consommation du circuit intégré complet qui commande R_{shunt} a été modélisée par une résistance pure $R_L = 20\text{ k}\Omega$. En pratique, le montage est composé :

- d'un système qui transforme l'onde alternative en tension quasi continue,
- d'un régulateur de tension et d'une électronique complexe.

Nous allons vérifier le principe de récupération de puissance par le secondaire au travers un montage classique redresseur mono alternance (1 diode) + condensateur de filtrage ($C_F = 1\text{ nF}$) pour délivrer une tension quasi continue.



On prélève la puissance reçue par le secondaire par un détecteur de crête diode + condensateur.

La netliste n'est que très peu modifiée : on commente la ligne R_L et on décommente les 3 suivantes.

```
*RL A 0 10k ;
Dpuce A K D1N4148
CF K 0 1n
RL K 0 30k ; modélise conso puce
```

Le même run que précédemment donne alors un bilan des puissances :

La puissance fournie par le signal carré à 125 kHz est de **66,4 mW**.

C'est quasiment la même valeur que le run précédent.

La base en consomme **11,8 mW** (par la résistance dans X_{demod}). Les autres composants « ne tirent pas » de puissance de cette source.

L'enroulement primaire dissipe **8,8 mW** dans r_p , et l'inductance primaire communique **45,5 mW**. Le secondaire reçoit ces **45,5 mW**.

Puis **11,2 mW** sont dissipées dans r_s .

La résistance de shunt dissipe **12,5mW**, la diode **0,6 mW**, et il reste **20,9 mW** dans $R_L = 30\text{ k}\Omega$.

C'est largement suffisant pour alimenter un circuit intégré.

X Values	Power
-AVGX(W(V125),400u)	66.443m
AVGX(W(Rp),400u)	8.7915m
AVGX(W(XDemod),400u)	11.816m
AVGX(W(L1),400u)	45.533m
-AVGX(W(L2),400u)	45.537m
AVGX(W(Rs),400u)	11.259m
AVGX(W(Rshunt),400u)	12.478m
AVGX(W(Dpuce),400u)	571.259u
AVGX(W(RL),400u)	20.924m

Bilan des puissances

Conclusion

Les principaux concepts (transmission en champ proche par couplage magnétique, modulation de charge, récupération d'énergie) ont été présentés.

L'expérience de début avec les 2 bobines formant une vraie manipulation dans le domaine RFID, a permis de présenter des simulations avec des valeurs numériques proches de la réalité. Néanmoins, pour des raisons de simulation avec la version Lite de Pspice, le schéma proposé est une version simplifiée de ce qui est présent dans les lecteurs RFID 125 kHz. Les résultats présentés sont très cohérents et représentatifs de la pratique.

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)